# Maintainer: Levente Polyak <anthraxx[at]archlinux[dot]org>
# Contributor: Alexandre Bouvier <contact AT amb DOT tf>
# Contributor: Ivan Shapovalov <intelfx@intelfx.name>
# Contributor: Tommaso Sardelli <lacapannadelloziotom AT gmail DOT com>
# Contributor: Philipp Joram <phijor AT t-online DOT de>

pkgname=python-axolotl-curve25519
_pkgver=0.4.1-2
pkgver=${_pkgver//-/.}
pkgrel=9
pkgdesc='Python wrapper for curve25519 library with ed25519 signatures'
url='https://github.com/tgalal/python-axolotl-curve25519'
arch=('loong64' 'x86_64')
license=('GPL3')
depends=('python' 'zlib' 'glibc')
makedepends=('python-setuptools')
source=(${pkgname}-${pkgver}.tar.gz::https://github.com/tgalal/python-axolotl-curve25519/archive/${_pkgver}.tar.gz)
sha256sums=('cc8de0c140831519e45598b082fdef0e939fd30930be442e38bdec13dc16aab5')
sha512sums=('cbc7c6caa47a9a811640c247a1be727d7b1b68bcdb4c5336e02b4d1eaf9fd2c57b7438b0da466a379a1c0f3f146756b9b7eea3c9b7945ce88478d4bf0b8a1e0d')

build() {
  echo "Building python..."
  cd ${pkgname}-${_pkgver}
  python setup.py build
}

package() {
  cd ${pkgname}-${_pkgver}
  python setup.py install --root="${pkgdir}" -O1 --skip-build
  install -Dm 644 README.md -t "${pkgdir}/usr/share/doc/${pkgname}"
}

# vim: ts=2 sw=2 et:
